
      PROGRAM DRYDEP                                                    

      IMPLICIT REAL(A-H,O-Z),INTEGER(I-N)        
      PARAMETER (NG=45)  ! number of gaseous species  
      PARAMETER (NLUC=26)   ! number of land use category
      REAL VDG(NLUC,NG)
      REAL Vdmax(NLUC,NG),Rns(NLUC,NG),Dratio(NG)
      INTEGER KDAY(12), KDAY1(12)                                       
      DATA KDAY/0,31,59,90,120,151,181,212,243,273,304,334/          
      DATA KDAY1/31,28,31,30,31,30,31,31,30,31,30,31/  
      character*500 line
      real Ra(NLUC),RST(NLUC),Ustar(NLUC), LAI_F(NLUC), LAI(26,15) 

C
C  calculate Vd for one site each time
C

C  initialize site-dependent information 
      GLAT=32.90289*3.14/180.    ! latitude for the site location, change accoridng
to your site 
      Z2  = 22.

      OPEN(55, file="MET.dat")
      read(55, '(A500)', ERR=475) line

475   continue

500   read(55, '(I5,3I3,12F17.10)', END=525) IYR, IMO, ID, IH,
     &  T2, Ts, U2, RH, SRAD, PREC, P0, SD, FCLD, RMOL, UstarObs,Wetness 

C  create output file
      OPEN (22,file='VD.dat')
      OPEN (33,file='METOUT.dat')
      OPEN (44,file='Res.dat')
      OPEN (11,file='Vdmax.dat')

C  find Julian day  
      JDAY=KDAY(IMO)+ID                                        
      if (IMO.GT.2.and.MOD(IYR,4).EQ.0 ) JDAY=JDAY+1.  

C -- Calculate solar zenith angle  
      hour=real(IH)
      DECLIN=ASIN(SIN(23.5*3.14159/180.)*
     &        SIN((JDAY-81.)*2.*3.14159/365.))
      SHORT1=SIN(GLAT)*SIN(DECLIN)
      SHORT2=COS(GLAT)*COS(DECLIN)
      COSZE=(HOUR-12)*3.14159/12.
      COSZEN=SHORT1+SHORT2*COS(COSZE)
      COSZEN= amax1(0.,COSZEN)



C Call GasVd to calculate Vd for 45 gaseous species

      call GasVd (Z2, u2, sd, t2, ts, srad, rh, fcld, prec,  
     &            COSZEN, P0, jday, RMOL, UstarObs,Wetness,           
     &            VDG, Ra, Ustar, Vdmax, RST,Rns,Dratio)            

      write(22,'(I4,3I3,90F10.5)') IYR,IMO,ID,IH,
     &     (VDG(4,J)*100,J=1,NG), (VDG(7,J)*100,J=1,NG)  ! m/s -> cm/s 
      write(11,'(I4,3I3,45F10.5)') IYR,IMO,ID,IH,
     &     (Vdmax(4,J)*100,J=1,NG) ! m/s -> cm/s

      write(33,'(I5,3I3,12F17.10)') IYR, IMO, ID, IH,
     &     T2,Ts,U2,RH, SRAD, PREC, P0, SD, FCLD, RMOL,UstarObs,Wetness
      write(44,'(I5,3I3,3F17.10,90F10.5,45F7.4)') IYR,IMO,ID,IH, Ra(4), 
     &     RST(4), RST(7),(100./Rns(4,J),J=1,NG),(100./Rns(7,J),J=1,NG),
     &     (Dratio(J),J=1,NG) 

      goto 500
525   continue

      close(11)
      close(22)
      close(33)
      close(44)

      STOP "!!!DONE!!!"   

      END                                                               

      SUBROUTINE init1 (V, K1) 
      real V(K1)
      do i=1,K1
        V(i)=0. 
      end do
      return
      end

      SUBROUTINE init2 (V, K1, K2) 
      real V(K1, K2)
      do i=1,K1
         do j=1,K2
           V(i,j)=0.
         end do 
      end do
      return
      end

      real function amin1 (x,y)
        amin1=x
        if (y.lt.x) amin1=y



      return
      end

      real function amax1 (x,y)
        amax1=x
        if (y.gt.x) amax1=y
      return
      end

      SUBROUTINE GasVd ( z2, u2, sd, t2, ts, srad, rh, fcld, prec,  
     &                   COSZEN, pmb, iday, RMOL, UstarObs,Wetness,
     &                   VDG, Ra, Ustar, Vdmax, RST, Rns,Dratio) 
C
C    PURPOSE: Calculate dry deposition velocities for 45 gas species
C                 including 31 species listed in Table 6 of Zhang et al. (2003) and
14 additional
C                 oVOC species listed in Table 1 of Wu et al. (2021) 
C                
C    References: 
C      Zhang et al., 2003. A revised parameterization for gaseous dry
C      deposition in air-quality model. Atmos. Chem. Phys., 3, 2067-2082,
C      https://doi.org/10.5194/acp-3-2067-2003
C
C      Wu et al., 2021. Extension of a gaseous dry deposition algorithm
C      to oxidized volatile organic compounds and hydrogen cyanide for
C      application in chemistry transport models. Geosci. Model Dev.
C      Discuss., 5:1-31, https://doi.org/10.5194/gmd-2021-41 
C
C                        leiming.zhang@canada.ca
C-----------------------------------------------------------------------
C        KEY   VARIABLES
C-----------------------------------------------------------------------
C  alpha    | Scaling factor based on SO2 (no unit) 
C  beta     | Scaling factor based on O3  (no unit) 
C  brs      | Constant for stomatal resistance(W/m2)
C  bvpd     | Constant for water vapor pressure deficit  (kPa^-1)                  

C  coszen   | Cosine of solar zenith angle         
C  fcld     | Cloud fraction (0.0-1.0)              
C  fland    | fraction of Land types (%)            
C  fsun     | fraction of sunlit leaves (0.0-LAI)  
C  iday     | Julian day                            
C  lai      | Leaf area index  (no unit)           
C  luc      | No. of land use category (26)        
C  mw       | molecular weight for gaseous species  
C           |   (g/mol)                             
C  ng       | No. of gaseous species dry deposited  
C  pardir   | visible beam radiation (W/m2)         
C  pardif   | diffuse visible radiation (W/m2)     
C  pmb      | Surface pressure (mb)              



C  prec     | hourly precipitation amount (mm/hour) 
C  psi1     | Constant for leaf water potential(Mpa)
C  psi2     | Constant for leaf water potential(Mpa)
C  fsnow    | Snow fraction  (0.0-1.0)              
C  ra       | Aerodynamic resistance (s/m)         
C  rac      | IN-canopy aerodynamic resistance (s/m)         
C  rb       | quasi-laminar resistance (s/m)      
C  rc       | total surface resistance (s/m)     
C  rcut     | cuticle resistance (s/m )              
C  rcutdo   | Dry cuticle resistance for O3 (s/m)  
C  rcutds   | Dry cuticle resistance for SO2 (s/m)
C  rcutwo   | Wet cuticle resistance for O3 (s/m)  
C  rg       | Ground  resistance (s/m )                
C  rgo      | Ground  resistance for O3 (s/m)  
C  rgs      | Ground  resistance for SO2 (s/m) 
C  rh       | relative humidity fraction (0.0-1.0)                
C  rm       | mesophyll resistance (s/m)           
C  rsmin    | minimum stomatal resistance (s/m)    
C  rst      | Stomatal resistance (s/m)           
C  sd       | Snow depth        (cm)             
C  sdmax    | Maximum snow depth over which snow  
C           |   fraction for leaves is 1 (cm )              
C  srad     | Solar irradiance (w/m2)          
C  t2       | Temperature  at first level (K)      
C  ts       | Surface temperature (K)             
C  tmin     | Minimum temperature for stomatal     
C           |    opening (C)                      
C  tmax     | Maxmum temperature for stomatal     
C           |    opening (C)                     
C  topt     | Optimum temperature for stomatal   
C           |    opening (C)                     
C  u2       | wind speed at reference height z2(m/s)
C  ustar    | friction velocity (m/s)               
C  VDF      | dry deposition velocity for one LUC   
C  VDG      | gaseous dry deposition velocity (m/s) 
C  wst      | fraction of stomatal closure under   
C           |      wet conditions (0.0-0.5)          
C  z0       | roughness length (m)                
C-----------------------------------------------------------------------

      IMPLICIT REAL(A-H,O-Z),INTEGER(I-N)        
      PARAMETER (NG=45)        ! NUMBER OF GAS SPECIES DRY DEPOSITED
      PARAMETER (LUC=26)       ! NUMBER OF LAND-USE CATEGORIES

      REAL Ra(LUC),RST(LUC)
C
C  paramaters 
C
      REAL Z01(LUC), Z02(LUC),  ustar(LUC)  



      REAL LAI(LUC,15),LAI_F(LUC)
C
C  paramaters for gaseous Vd submoudle 
C
      REAL VDG(LUC,NG), ALPHA(NG),BETA(NG),RM(NG),MW(NG)
      REAL Rac1(LUC),   Rac2(LUC), RcutdO(LUC), RcutwO(LUC), 
     &   RcutdS(LUC),   RgS(LUC),  RgO(LUC),  SDmax(LUC),  
     &    Tmin(LUC),   Tmax(LUC),   TOPT(LUC),   BVPD(LUC),  
     &    PSI1(LUC),   PSI2(LUC),  RSmin(LUC),    BRS(LUC)
      LOGICAL is_dew, is_rain
      REAL Vdmax(LUC,NG), Rns(LUC,NG),Dratio(NG)

C
C  external functions 
C
      external amin1, amax1 

C
C   Surface Roughness Length [m].
C   Z01 and Z02 are minimum and maximum z0 for each LUC. 
C
      DATA   Z01             /
     &  0.0 ,  0.01,  0.0 ,  0.9 ,  2.0 , 
     &  0.4 ,  0.4 ,  2.5 ,  0.6 ,  0.2 ,
     &  0.05,  0.2 ,  0.04,  0.02,  0.02,
     &  0.02,  0.02,  0.02,  0.02,  0.05,
     &  1.0 ,  0.03,  0.1 ,  0.04,  0.6 , 
     &  0.6    /
      DATA   Z02             /
     &  0.0 ,  0.01,  0.0 ,  0.9 ,  2.0 , 
     &  0.9 ,  1.0 ,  2.5 ,  0.6 ,  0.2 ,
     &  0.2 ,  0.2 ,  0.04,  0.1 ,  0.1 ,
     &  0.1 ,  0.1 ,  0.1 ,  0.2 ,  0.05,
     &  1.0 ,  0.03,  0.1 ,  0.04,  0.9 , 
     &  0.9    /
C
C   In-canopy aerodynamic resistance  [s/m].
C   Rac1 and Rac2 are minimum and maximum Rac0 for each LUC. 
C
      DATA   Rac1            /
     &  0   ,  0   ,  0   ,  100 ,  250 , 
     &  60  ,   60 ,  300 ,  100 ,  60  ,
     &  20  ,  40  ,  20  ,  10  ,  10  ,
     &  10  ,  10  ,  10  ,  10  ,  20  ,
     &  40  ,  0   ,  20  ,  0   ,  100 ,
     &  100    /
      DATA   Rac2            /
     &  0   ,  0   ,  0   ,  100 ,  250 , 
     &  100 ,  100 ,  300 ,  100 ,  60  ,
     &  60  ,  40  ,  20  ,  40  ,  40  ,



     &  40  ,  40  ,  50  ,  40  ,  20  ,
     &  40  ,  0   ,  20  ,  0   ,  100 ,
     &  100    /
C
C   Dry and wet cuticle resistance for O3  [s/m].
C
      DATA   RcutdO          /
     & -999 , -999 , -999 , 4000 , 6000 , 
     & 4000 , 6000 , 6000 , 8000 , 6000 ,
     & 5000 , 5000 , 4000 , 4000 , 4000 ,
     & 4000 , 4000 , 5000 , 5000 , 4000 ,
     & 6000 , 8000 , 5000 , -999 , 4000 ,
     & 4000    /
      DATA   RcutwO          /
     & -999 , -999 , -999 ,  200 ,  400 , 
     &  200 ,  400 ,  400 ,  400 ,  400 ,
     &  300 ,  300 ,  200 ,  200 ,  200 ,
     &  200 ,  200 ,  300 ,  300 ,  200 ,
     &  400 ,  400 ,  300 , -999 ,  200 ,
     &  200    /
C
C   Ground resistance for O3  [s/m].
C
      DATA   RgO            /
     & 2000 , 2000 , 2000 ,  200 ,  200 , 
     &  200 ,  200 ,  200 ,  200 ,  200 ,
     &  200 ,  200 ,  200 ,  200 ,  200 ,
     &  200 ,  200 ,  200 ,  200 ,  500 ,
     &  500 ,  500 ,  500 ,  500 ,  200 ,
     &  200    /
C
C   Dry cuticle resistance for SO2  [s/m].
C
      DATA   RcutdS          /
     & -999 , -999 , -999 , 2000 , 2500 , 
     & 2000 , 2500 , 2500 , 6000 , 2000 ,
     & 2000 , 2000 , 1000 , 1000 , 1500 ,
     & 1500 , 2000 , 2000 , 2000 , 2000 ,
     & 4000 , 2000 , 1500 , -999 , 2500 ,
     & 2500    /
C
C   Ground resistance for SO2  [s/m].
C
      DATA   RgS            /
     &   20 ,   70 ,  20  ,  200 ,  100 , 
     &  200 ,  200 ,  100 ,  300 ,  200 ,
     &  200 ,  200 ,  200 ,  200 ,  200 ,
     &   50 ,  200 ,  200 ,  200 ,   50 ,
     &  300 ,  300 ,   50 ,  700 ,  200 ,
     &  200    /



C
C   Stomatal resistance related parameters. 
C   In sequence: rsmin, brs, tmin, tmax, topt, bvpd, psi1, psi2
C
      DATA   rsmin            /
     & -999 , -999 , -999 ,  250 ,  150 ,
     &  250 ,   150,  150 ,  250 ,  150 ,
     &  150 ,  250 ,  150 ,  100 ,  120 , 
     &  120 ,  120 ,  250 ,  125 ,  150 ,
     &  200 ,  150 ,  150 , -999 ,  150 , 
     &  150   /
      DATA   brs              /
     & -999 , -999 , -999 ,   44 ,   40 , 
     &   44 ,   43 ,   40 ,   44 ,   40 , 
     &   44 ,   44 ,   50 ,   20 ,   40 , 
     &   40 ,   50 ,   65 ,   65 ,   40 , 
     &   42 ,   25 ,   40 , -999 ,   44 , 
     &   43   /
      DATA   tmin             /
     & -999 , -999 , -999 ,   -5 ,    0 , 
     &   -5 ,    0 ,    0 ,    0 ,    0 , 
     &   -5 ,    0 ,    5 ,    5 ,    5 , 
     &    5 ,    5 ,    5 ,   10 ,    5 , 
     &    0 ,   -5 ,    0 , -999 ,   -3 , 
     &    0   /
      DATA   tmax             /
     & -999 , -999 , -999 ,   40 ,   45 , 
     &   40 ,   45 ,   45 ,   45 ,   45 , 
     &   40 ,   45 ,   40 ,   45 ,   45 , 
     &   45 ,   45 ,   45 ,   45 ,   45 , 
     &   45 ,   40 ,   45 , -999 ,   42 , 
     &   45   /
      DATA   topt             /
     & -999 , -999 , -999 ,   15 ,   30 , 
     &   15 ,   27 ,   30 ,   25 ,   30 , 
     &   15 ,   25 ,   30 ,   25 ,   27 , 
     &   27 ,   25 ,   25 ,   30 ,   25 , 
     &   22 ,   20 ,   20 , -999 ,   21 , 
     &   25   /
      DATA   bvpd             /
     & -999 , -999 , -999 ,  0.31,  0.27, 
     &  0.31,  0.36,  0.27,  0.31,  0.27,  
     &  0.27,  0.27,  0.0 ,  0.0 ,  0.0 ,  
     &  0.0 ,  0.0 ,  0.0 ,  0.0 ,  0.0 ,  
     &  0.31,  0.24,  0.27, -999 ,  0.34, 
     &  0.31   /
      DATA   psi1            /
     & -999 , -999 , -999 , -2.0 , -1.0 , 
     & -2.0 , -1.9 , -1.0 , -1.0 , -2.0 ,
     & -2.0 , -2.0 , -1.5 , -1.5 , -1.5 ,



     & -1.5 , -1.5 , -1.5 , -1.5 , -1.5 ,
     & -1.5 ,    0 , -1.5 , -999 , -2.0 ,
     & -2.0    /
      DATA   psi2            /
     & -999 , -999 , -999 , -2.5 , -5.0 , 
     & -2.5 , -2.5 , -5.0 , -4.0 , -4.0 ,
     & -4.0 , -3.5 , -2.5 , -2.5 , -2.5 ,
     & -2.5 , -2.5 , -2.5 , -2.5 , -2.5 ,
     & -3.0 , -1.5 , -2.5 , -999 , -2.5 ,
     & -3.0    /
C
C Leaf area index at the beginning of each month (im=1,13), 
C  minimum LAI (im=14) and maximum LAI (im=15).
C Values of LAI are from GEM, provided by Stephane Belair and Judy St-James,
C with modifications for urban.
C
      DATA (LAI(6,im), im = 1, 15)/ 
     & 0.1  , 0.1  , 0.5  , 1.0  , 2.0  , 
     & 4.7  , 4.7  , 5.0  , 4.0  , 2.0  , 
     & 1.0  , 0.1  , 0.1  , 0.1  , 5.0  /
      DATA (LAI(7,im), im = 1, 15)/ 
     & 0.1  , 0.1  , 0.5  , 1.0  , 2.0  , 
     & 4.7  , 4.7  , 5.0  , 4.0  , 2.0  , 
     & 1.0  , 0.1  , 0.1  , 0.1  , 5.0  / 
      DATA (LAI(11,im), im = 1, 15)/ 
     & 0.5  , 0.5  , 1.0  , 1.0  , 1.5  , 
     & 2.0  , 3.0  , 3.0  , 2.0  , 1.5  , 
     & 1.0  , 0.5  , 0.5  , 0.5  , 3.0  /
      DATA (LAI(14,im), im = 1, 15)/ 
     & 0.5  , 0.5  , 0.5  , 0.5  , 0.5  , 
     & 0.5  , 1.0  , 2.0  , 2.0  , 1.5  , 
     & 1.0  , 1.0  , 0.5  , 0.5  , 2.0  /
      DATA (LAI(15,im), im = 1, 15)/ 
     & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  , 
     & 2.0  , 3.0  , 3.5  , 4.0  , 0.1  , 
     & 0.1  , 0.1  , 0.1  , 0.1  , 4.0  /
      DATA (LAI(16,im), im = 1, 15)/ 
     & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  , 
     & 2.5  , 4.0  , 5.0  , 6.0  , 0.1  , 
     & 0.1  , 0.1  , 0.1  , 0.1  , 6.0  /
      DATA (LAI(17,im), im = 1, 15)/ 
     & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  , 
     & 3.0  , 4.0  , 4.5  , 5.0  , 0.1  , 
     & 0.1  , 0.1  , 0.1  , 0.1  , 5.0  /
      DATA (LAI(18,im), im = 1, 15)/ 
     & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  , 
     & 2.0  , 3.0  , 3.5  , 4.0  , 0.1  , 
     & 0.1  , 0.1  , 0.1  , 0.1  , 4.0  /
      DATA (LAI(19,im), im = 1, 15)/ 
     & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  , 



     & 3.0  , 4.0  , 4.5  , 5.0  , 0.1  , 
     & 0.1  , 0.1  , 0.1  , 0.1  , 5.0  /
      DATA (LAI(21,im), im = 1, 15)/ 
     & 0.1  , 0.1  , 0.1  , 0.1  , 0.5  , 
     & 1.0  , 1.0  , 1.0  , 1.0  , 1.0  , 
     & 0.4  , 0.1  , 0.1  , 0.1  , 1.0  /
      DATA (LAI(22,im), im = 1, 15)/ 
     & 1.0  , 1.0  , 0.5  , 0.1  , 0.1  , 
     & 0.1  , 0.1  , 1.0  , 2.0  , 1.5  , 
     & 1.5  , 1.0  , 1.0  , 0.1  , 2.0  /
      DATA (LAI(25,im), im = 1, 15)/ 
     & 3.0  , 3.0  , 3.0  , 4.0  , 4.5  , 
     & 5.0  , 5.0  , 5.0  , 4.0  , 3.0  , 
     & 3.0  , 3.0  , 3.0  , 3.0  , 5.0  /
      DATA (LAI(26,im), im = 1, 15)/ 
     & 3.0  , 3.0  , 3.0  , 4.0  , 4.5  , 
     & 5.0  , 5.0  , 5.0  , 4.0  , 3.0  , 
     & 3.0  , 3.0  , 3.0  , 3.0  , 5.0  /
C 
C  Gas Properties (Mesophyll resistance RM, scaling factors ALPHA and BETA, 
C  and molecular weight) for a total of 45 species in this sequence: 
C
C    1 SO2
C    2 H2SO4
C    3 NO2  
C    4 O3
C    5 H2O2
C    6 HNO3 
C    7 HONO  
C    8 HNO4
C    9 NH3
C    10 PAN   
C    11 PPN   
C    12 APAN  
C    13 MPAN  
C    14 HCHO  
C    15 MCHO  
C    16 PALD  
C    17 C4A   
C    18 C7A   
C    19 ACHO  
C    20 MVK   
C    21 MACR  
C    22 MGLY
C    23 MOH   
C    24 ETOH  
C    25 POH   
C    26 CRES
C    27 FORM  
C    28 ACAC  



C    29 ROOH  
C    30 ONIT  
C    31 INIT  
C    32 HCN
C    33 HMHP
C    34 HAC
C    35 PAA
C    36 HDC4
C    37 DHC4
C    38 HPALD
C    39 ISOPOOH 
C    40 IEPOX 
C    41 PROPNN
C    42 ISOPN
C    43 MACN/MVKN 
C    44 INP
C    45 MTNP

      DATA   RM            /
     & 0.   , 0.   , 0.   ,  0.  ,  0.  , 
     & 0.   , 0.   , 0.   ,  0.  ,  0.  , 
     & 0.   , 0.   , 0.   ,  0.  ,  100., 
     & 100. , 100. , 100. ,  100.,  0.  , 
     & 100. , 0.   , 0.   ,  0.  ,  0.  , 
     & 0.   , 0.   , 0.   ,  0.  ,  100.,
     & 100. , 100. , 0.   ,  0.  ,  0.  ,
     & 0.   , 0.   , 0.   ,  0.  ,  0.  ,
     & 0.   , 0.   , 0.   ,  0.  ,  0./ 

      DATA   ALPHA          /
     & 1.   , 1.   , 0.   ,  0.  ,  1.  , 
     & 10.  , 2.   , 5.   ,  1.  ,  0.  , 
     & 0.   , 0.   , 0.   ,  0.8 ,  0.  , 
     & 0.   , 0.   , 0.   ,  0.  ,  0.  , 
     & 0.   , 0.01 , 0.6  ,  0.6 ,  0.4 , 
     & 0.01 , 2.   , 1.5  ,  0.1 ,  0.  , 
     & 0.   , 0.   , 5.   ,  1.5  ,  2.  ,
     & 1.   , 2.   , 1.5  ,  5.   ,  5.  ,
     & 1.5  ,  1.5 , 1.5  ,  1.5  ,  1.5  /

      DATA   BETA          /
     & 0.   , 1.   , 0.8  ,  1.  ,  1.  , 
     & 10.  , 2.   , 5.   ,  0.0 ,  0.6 , 
     & 0.6  , 0.8  , 0.3  ,  0.2 ,  0.05, 
     & 0.05 , 0.05 , 0.05 ,  0.05,  0.05, 
     & 0.05 , 0.   , 0.1  ,  0.  ,  0.  , 
     & 0.   , 0.2  , 0.   ,  0.8 ,  0.5 ,
     & 0.5  , 0.1  , 1.   ,  1.  ,  1. ,
     & 0.2  , 0.2  , 1.   ,  0.2 ,  0.2 ,
     & 1.   , 1.   , 1.   ,  1.  ,  1./



      DATA   MW          /
     & 64.  , 98.  , 46.  ,  48. ,  34. , 
     & 63.  , 47.  , 79.  ,  17. ,  121.,
     & 135. , 183. , 147. ,  30. ,  44. ,
     & 58.  , 72.  , 128. ,  106.,  70. ,
     & 70.  , 72.  , 32.  ,  46. ,  60. ,
     & 104. , 46.  , 60.  ,  48. ,  77. ,
     & 147. , 27.  , 64.  ,  74. ,  76. ,
     & 102. , 104. , 116. ,  118.,  118., 
     & 119. , 147. , 149. ,  163.,  231./
C
C  Maximum snow depth over which snow fraction for leaves is 1.0
C  Snow fraction for ground is treated 2 times of that for leaves 
C
      DATA SDMAX  /
     &              9999. , 1.0   , 9999. ,  200. ,  400. ,
     &               200. , 200.  ,  400. ,  200. ,   50. ,
     &                50. ,  50.  ,   5.  ,   20. ,   10. , 
     &                10. ,  10.  ,  10.  ,   10. ,   10. , 
     &                50. ,   2.  ,  10.  ,    2. ,  200. , 
     &               200.        / 

C ---  parameters for air dynamic properties
      DATA ROAROW/1.19/

C
C --- Define the function for saturation vapor pressure (mb) 
C
      ES(TEMP) = 6.108*EXP(17.27*(TEMP - 273.16)/(TEMP - 35.86))
C 
C   Some constants 
C
      dair=0.369*29.+6.29  
      dh2o=0.369*18.+6.29

C   Initialize Leaf Area Index for LUC with constant LAI values
      DO im=1, 15 
      LAI(1,im)=0.
      LAI(2,im)=0.
      LAI(3,im)=0.
      LAI(4,im)=4.7
      LAI(5,im)=6.
      LAI(8,im)=6.
      LAI(9,im)=4.
      LAI(10,im)=3.
      LAI(12,im)=3.
      LAI(13,im)=1.
      LAI(20,im)=1.
      LAI(23,im)=4.



      LAI(24,im)=0.
      END DO

      VDG=0.
      Vdmax=0.
      Rns=0.

C
C Loop 200 for LUC  
C
      DO 200 I=1,LUC

C
C interpolate LAI
C
           IM = INT(iday / 30.5 ) + 1
           iday_M =iday - INT((IM-1)*30.5+0.5) 
           IF (iday_M.EQ.0) THEN
           IM=IM-1
           iday_M =iday - (IM-1)*30.5
           END IF

           LAI_F(I)  = LAI(I,IM) 
     &          + iday_M / 30.5 * (LAI(I,IM+1)-LAI(I,IM))

      Z0_F  = (0.23-LAI_F(I)**0.25/10-(2-1)/67.)*10.
      DDD   = (0.05+LAI_F(I)**0.20/2.+(2-1)/20.)*10.
      ZL=(Z2-DDD)*RMOL
      USTAR(I)= UstarObs

C
C     Aerodynamic resistance above canopy
C
      IF(ZL.GE.0.) THEN
          Ra(I)=(.74*ALOG((Z2-DDD)/Z0_F)+4.7*ZL)/0.4/USTAR(I) 
      ELSE
          Ra(I)=0.74/0.4/USTAR(I)*(ALOG((Z2-DDD)/Z0_F)-
     &          2*ALOG((1+SQRT(1-9.*ZL))*0.5))
      ENDIF

      Ra(I)=amax1(Ra(I),1.0)

      if (I.EQ.1.OR.I.EQ.3) THEN
          Ra(I)=amin1(Ra(I),2000.)
      else
          Ra(I)=amin1(Ra(I),1000.)
      end if 

C
C --- STOMATAL RESISTANCE FOR WATER VAPOR ONLY. STEPS FOR CALCULATING:



C     1. Calculate direct and diffuse PAR from solar radiation
C     2. Calculate sunlit and shaded leaf area, PAR for sunlit and shaded leafs 
C     3. Calculate stomatal conductance 
C     4. Calculate stomatal resistance for water vapor
C
C
C --  Only calculate stomatal resistance if there is solar radiation, 
C     leaf area index is not zero, and within reasonable temperature range
C
      IF ( SRAD.GE.0.1             .AND.
     &     TS.LT.(Tmax(I)+273.15)  .AND. 
     &     TS.GT.(Tmin(I)+273.15)  .AND.  
     &     LAI_F(I).GT.0.001       .AND.
     &     COSZEN.GT.0.001       ) THEN

C --  Calculate direct and diffuse PAR from solar radiation and solar zenith angle

      RDU=600.*EXP(-0.185/COSZEN)*COSZEN
      RDV=0.4*(600.-RDU)*COSZEN
      WW=-ALOG(COSZEN)/2.302585  
      WW=-1.195+0.4459*WW-0.0345*WW**2
      WW=1320*10**WW
      RDM=(720.*EXP(-0.06/COSZEN)-WW)*COSZEN
      RDN=0.6*(720-RDM-WW)*COSZEN
      RV=amax1(0.1,RDU+RDV) 
      RN=amax1(0.01,RDM+RDN) 
      RATIO=amin1(0.9,SRAD/(RV+RN))
      SV=RATIO*RV                            ! Total PAR 
      FV=amin1(0.99, (0.9-RATIO)/0.7)  
      FV=amax1(0.01,RDU/RV*(1.0-FV**0.6667))  !fraction of PAR in the direct beam 
      PARDIR=FV*SV                           ! PAR from direct radiation 
      PARDIF=SV-PARDIR                       ! PAR from diffuse radiation 
c  
C -- Calculate sunlit and shaded leaf area, PAR for sunlit and shaded leaves 
C
      IF (LAI_F(I).GT.2.5.AND.SRAD.GT.200.) THEN
      PSHAD=PARDIF*EXP(-0.5*LAI_F(I)**0.8)
     & +0.07*PARDIR*(1.1-0.1*LAI_F(I))*EXP(-COSZEN)
      PSUN=PARDIR**0.8*.5/COSZEN+PSHAD
      ELSE
      PSHAD=PARDIF*EXP(-0.5*LAI_F(I)**0.7)
     & +0.07*PARDIR*(1.1-0.1*LAI_F(I))*EXP(-COSZEN)
      PSUN=PARDIR*.5/COSZEN+PSHAD
      END IF
      RSHAD=RSmin(I)+BRS(I)*RSMIN(I)/PSHAD 
      RSUN=RSmin(I)+BRS(I)*RSMIN(I)/PSUN
      GSHAD=1./RSHAD
      GSUN=1./RSUN
      FSUN=2*COSZEN*(1.-EXP(-0.5*LAI_F(I)/COSZEN))  ! Sunlit leaf area 
      FSHAD=LAI_F(I)-FSUN                    ! Shaded leaf area 



C -- Stomatal conductance before including effects of temperature, 
C                   vapor pressure defict and water stress.

      GSPAR=FSUN*GSUN+FSHAD*GSHAD     

C --  function for temperature effect
      T=TS-273.15      
      BT=(Tmax(I)-TOPT(I))/(TOPT(I)-Tmin(I))
      GT=(Tmax(I)-T)/(TMAX(I)-TOPT(I))
      GT=GT**BT
      GT=GT*(T-Tmin(I))/(TOPT(I)-TMIN(I))  
C --  function for vapor pressure deficit 
      D0= ES(TS)*(1.- RH*1.05)/10.           !kPa 
      GD=1.-BVPD(I)*D0
C --  function for water stress 
      PSI=(-0.72-0.0013*SRAD)
      GW=(PSI-PSI2(I))/(PSI1(I)-PSI2(I))
      IF (GW.GT.1.0) GW=1.0
      IF (GW.LT.0.1) GW=0.1
      IF (GD.GT.1.0) GD=1.0
      IF (GD.LT.0.1) GD=0.1
C --  Stomatal resistance for water vapor 
      RST(I)=1.0/(GSPAR*GT*GD*GW)    

      ELSE
      RST(I)=99999.9
      END IF
C
c   Decide if dew or rain occurs.
C
      IF (FCLD.LT.0.25) THEN
        Coedew=0.3  
      ELSE  IF (FCLD.GE.0.25.AND.FCLD.LT.0.75) THEN 
        Coedew=0.2  
      ELSE
        Coedew=0.1  
      END IF
         DQ=0.622/1000. * ES(TS)*(1.- RH)*1000.    ! unit g/kg 
         DQ=amax1(0.0001,DQ)
         USMIN=1.5/DQ*Coedew  

      is_rain = .false.
      is_dew  = .false.
      IF (T2.GT.273.15) THEN
         if (PREC.GT.0.20) then
            is_rain = .true.
         elseif (Wetness.GT.0.8) then
            is_dew  = .true.
         endif



      ENDIF

C
C   Decide fraction of stomatal blocking due to wet conditions 
C 
      Wst=0.
      if ((is_dew.or.is_rain).and.SRAD.GT.200.) then
      Wst=(SRAD-200.)/800.
      Wst=amin1(Wst, 0.5) 
      end if 
C
C -- In-canopy aerodynamic resistance 
C
        Rac = Rac1(I)+(LAI_F(I)-LAI(I,14))/(LAI(I,15)-LAI(I,14)+1.E-10)
     &           *(Rac2(I)-Rac1(I)) 
        Rac = Rac*LAI_F(I)**0.25/USTAR(I)/USTAR(I) 
C
C -- Ground resistance for O3 
C
      IF (I.GE.4.AND.TS.LT.272.15) THEN 
        RgO_F = amin1( RgO(I)*2., RgO(I) * exp(0.2*(272.15-TS))) 
      ELSE
        RgO_F = RgO(I)
      END IF
C
C -- Ground resistance for SO2 
C
      IF (I.EQ.2) THEN 
        RgS_F = AMIN1(RgS(I)*(275.15-TS), 500.) 
        RgS_F = AMAX1(RgS(I), 100.) 
      ELSE IF (I.GE.4.AND.is_rain) THEN
        RgS_F = 50. 
      ELSE IF (I.GE.4.AND.is_dew) THEN
        RgS_F = 100. 
      ELSE IF (I.GE.4.AND.TS.LT.272.15) THEN
        RgS_F = amin1( RgS(I)*2., RgS(I) * exp(0.2*(272.15-TS))) 
      ELSE 
        RgS_F =  RgS(I)
      END IF
C
C -- Cuticle resistance for O3 AND SO2  
C
      IF (RcutdO(I).LE.-1) THEN 
        RcutO_F = 1.E25 
        RcutS_F = 1.E25 
      ELSE IF (is_rain) THEN
        RcutO_F = RcutwO(I)/LAI_F(I)**0.5/USTAR(I) 
        RcutS_F = 50./LAI_F(I)**0.5/USTAR(I) 
        RcutS_F = MAX (RcutS_F, 20.) 
      ELSE IF (is_dew) THEN



        RcutO_F = RcutwO(I)/LAI_F(I)**0.5/USTAR(I) 
        RcutS_F = 100./LAI_F(I)**0.5/USTAR(I) 
        RcutS_F = MAX (RcutS_F, 20.) 
      ELSE IF (TS.LT.272.15) THEN
        RcutO_F = RcutdO(I)/exp(3.*RH)/LAI_F(I)**0.25/USTAR(I) 
        RcutS_F = RcutdS(I)/exp(3.*RH)/LAI_F(I)**0.25/USTAR(I) 
        RcutO_F = amin1( RcutO_F*2., RcutO_F * exp(0.2*(272.15-TS))) 
        RcutS_F = amin1( RcutS_F*2., RcutS_F * exp(0.2*(272.15-TS))) 
        RcutO_F = MAX (RcutO_F,100.) 
        RcutS_F = MAX (RcutS_F,100.) 
      ELSE 
        RcutO_F = RcutdO(I)/exp(3.*RH)/LAI_F(I)**0.25/USTAR(I) 
        RcutS_F = RcutdS(I)/exp(3.*RH)/LAI_F(I)**0.25/USTAR(I) 
        RcutO_F = MAX (RcutO_F,100.) 
        RcutS_F = MAX (RcutS_F,100.) 
      END IF
C
C If snow occurs, Rg and Rcut are adjusted by snow cover fraction
C
        fsnow= sd/sdmax(i)
        fsnow= amin1(1.0, fsnow)   !snow cover fraction for leaves 
      If (fsnow.GT.0.0001.and.I.GE.4) THEN
        RsnowS= AMIN1(70.*(275.15-TS), 500.) 
        RsnowS= AMAX1(RSnowS, 100.) 
        RcutS_F=1.0/((1.-fsnow)/RcutS_F+fsnow/RsnowS) 
        RcutO_F=1.0/((1.-fsnow)/RcutO_F+fsnow/2000.) 
        fsnow= amin1(1.0, fsnow*2.)   !snow cover fraction for ground 
        RgS_F=1.0/((1.-fsnow)/RgS_F+fsnow/RsnowS) 
        RgO_F=1.0/((1.-fsnow)/RgO_F+fsnow/2000.) 
      END IF

C
C Loop 100 for gas species 
C
      DO 100 J=1,NG
C
C -- Calculate diffusivity for each gas species
C
        dgas=0.369*MW(J)+6.29  
        DI=0.001*TS**1.75*SQRT((29.+MW(J))/MW(J)/29.)      
        DI=DI/1.0/(dair**0.3333+dgas**0.3333)**2   
        VI=145.8*1.E-4*(TS*0.5+T2*0.5)**1.5/
     &                  (TS*0.5+T2*0.5+110.4)
        VI=VI/ROAROW
C
C -- Calculate quasi-laminar resistance 
C
        Rb  =5./USTAR(I)*(VI/DI)**.666667   
C
C -- Calculate stomatal resistance for each species from the ratio of



C       diffusity of water vapor to the gas species
C
        DVh2o=0.001*TS**1.75*SQRT((29.+18.)/29./18.)
        DVh2o=DVh2o/(dair**0.3333+dh2o**0.3333)**2

        RS=RST(I)*DVh2o/DI+RM(J) 
C
C -- Scale cuticle and ground resistances for each species 
C
        Rcut = 1./(ALPHA(J)/RcutS_F+BETA(J)/RcutO_F)
        Rg = 1./(ALPHA(J)/RgS_F+BETA(J)/RgO_F)
C
C -- Calculate total surface resistance  
C
        Rc = (1.-Wst)/Rs+1./(Rac+Rg)+1./Rcut 
        Rc=amax1(0.,1./Rc)

C
C -- Deposition velocity 
C
        VDG(I,J) = 1./(RA(I)+RB+RC)  
        Vdmax(I,J) = 1./(RA(I)+RB)
        Rns(I,J) = 1./(1./(Rac+Rg)+1./Rcut)
        Dratio(J) = DVh2o/DI

100   CONTINUE   ! end of gaseous species 
200   CONTINUE   ! end of LUC 

      RETURN

      END SUBROUTINE GasVd


